

NAG Library Routine Document

S30FAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of **bold italicised** terms and other implementation-dependent details.

1 Purpose

S30FAF computes the price of a standard barrier option.

2 Specification

```

SUBROUTINE S30FAF(CALPUT, TYPE, M, N, X, S, H, K, T, SIGMA, R, Q, P,
1 LDP, IFAIL)
INTEGER M, N, LDP, IFAIL
double precision X(M), S, H, K, T(N), SIGMA, R, Q, P(LDP,N)
CHARACTER*1 CALPUT
CHARACTER*2 TYPE

```

3 Description

S30FAF computes the price of a standard barrier option, where the exercise, for a given strike price, X , depends on the underlying asset price, S , reaching or crossing a specified barrier level, H . Barrier options of type **In** only become active (are knocked in) if the underlying asset price attains the pre-determined barrier level during the lifetime of the contract. Those of type **Out** start active and are knocked out if the underlying asset price attains the barrier level during the lifetime of the contract. A cash rebate, K , may be paid if the option is inactive at expiration. The option may also be described as **Up** (the underlying price starts below the barrier level) or **Down** (the underlying price starts above the barrier level). This gives the following options which can be specified as put or call contracts.

Down-and-In: the option starts inactive with the underlying asset price above the barrier level. It is knocked in if the underlying price moves down to hit the barrier level before expiration.

Down-and-Out: the option starts active with the underlying asset price above the barrier level. It is knocked out if the underlying price moves down to hit the barrier level before expiration.

Up-and-In: the option starts inactive with the underlying asset price below the barrier level. It is knocked in if the underlying price moves up to hit the barrier level before expiration.

Up-and-Out: the option starts active with the underlying asset price below the barrier level. It is knocked out if the underlying price moves up to hit the barrier level before expiration.

The payoff is $\max(S - X, 0)$ for a call or $\max(X - S, 0)$ for a put, if the option is active at expiration, otherwise it may pay a pre-specified cash rebate, K . Following Haug (2007), the prices of the various standard barrier options can be written as shown below. The volatility, σ , risk-free interest rate, r , and annualised dividend yield, q , are constants. The integer parameters, j and k , take the values ± 1 , depending on the type of barrier.

$$\begin{aligned}
 A &= jSe^{-qT}\Phi(jx_1) - jXe^{-rT}\Phi(j[x_1 - \sigma\sqrt{T}]) \\
 B &= jSe^{-qT}\Phi(jx_2) - jXe^{-rT}\Phi(j[x_2 - \sigma\sqrt{T}]) \\
 C &= jSe^{-qT}\left(\frac{H}{S}\right)^{2(\mu+1)}\Phi(ky_1) - jXe^{-rT}\left(\frac{H}{S}\right)^{2\mu}\Phi(k[y_1 - \sigma\sqrt{T}]) \\
 D &= jSe^{-qT}\left(\frac{H}{S}\right)^{2(\mu+1)}\Phi(ky_2) - jXe^{-rT}\left(\frac{H}{S}\right)^{2\mu}\Phi(k[y_2 - \sigma\sqrt{T}]) \\
 E &= Ke^{-rT}\left\{\Phi(k[x_2 - \sigma\sqrt{T}]) - \left(\frac{H}{S}\right)^{2\mu}\Phi(k[y_2 - \sigma\sqrt{T}])\right\} \\
 F &= K\left\{\left(\frac{H}{S}\right)^{\mu+\lambda}\Phi(kz) + \left(\frac{H}{S}\right)^{\mu-\lambda}\Phi(k[z - \sigma\sqrt{T}])\right\}
 \end{aligned}$$

with

$$\begin{aligned}
x_1 &= \frac{\ln(S/X)}{\sigma\sqrt{T}} + (1 + \mu)\sigma\sqrt{T} \\
x_2 &= \frac{\ln(S/H)}{\sigma\sqrt{T}} + (1 + \mu)\sigma\sqrt{T} \\
y_1 &= \frac{\ln(H^2/(SX))}{\sigma\sqrt{T}} + (1 + \mu)\sigma\sqrt{T} \\
y_2 &= \frac{\ln(H/S)}{\sigma\sqrt{T}} + (1 + \mu)\sigma\sqrt{T} \\
z &= \frac{\ln(H/S)}{\sigma\sqrt{T}} + \lambda\sigma\sqrt{T} \\
\mu &= \frac{r-q-\sigma^2/2}{\sigma^2} \\
\lambda &= \sqrt{\mu^2 + \frac{2r}{\sigma^2}}
\end{aligned}$$

and where Φ denotes the cumulative Normal distribution function,

$$\Phi(x) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^x \exp(-y^2/2) dy.$$

Down-and-In ($S > H$):

When $X \geq H$, with $j = k = 1$,

$$P_{\text{call}} = C + E$$

and with $j = -1$, $k = 1$

$$P_{\text{put}} = B - C + D + E$$

When $X < H$, with $j = k = 1$

$$P_{\text{call}} = A - B + D + E$$

and with $j = -1$, $k = 1$

$$P_{\text{put}} = A + E$$

Down-and-Out ($S > H$):

When $X \geq H$, with $j = k = 1$,

$$P_{\text{call}} = A - C + F$$

and with $j = -1$, $k = 1$

$$P_{\text{put}} = A - B + C - D + F$$

When $X < H$, with $j = k = 1$,

$$P_{\text{call}} = B - D + F$$

and with $j = -1$, $k = 1$

$$P_{\text{put}} = F$$

Up-and-In ($S < H$):

When $X \geq H$, with $j = 1$, $k = -1$,

$$P_{\text{call}} = A + E$$

and with $j = k = -1$,

$$P_{\text{put}} = A - B + D + E$$

When $X < H$, with $j = 1$, $k = -1$,

$$P_{\text{call}} = B - C + D + E$$

and with $j = k = -1$,

$$P_{\text{put}} = C + E$$

Up-and-Out ($S < H$):

When $X \geq H$, with $j = 1$, $k = -1$,

$$P_{\text{call}} = F$$

and with $j = k = -1$,

$$P_{\text{put}} = B - D + F$$

When $X < H$, with $j = 1$, $k = -1$,

$$P_{\text{call}} = A - B + C - D + F$$

and with $j = k = -1$,

$$P_{\text{put}} = A - C + F$$

4 References

Haug E G (2007) *The Complete Guide to Option Pricing Formulas* (2nd Edition) McGraw-Hill

5 Parameters

- 1: CALPUT – CHARACTER*1 *Input*
On entry: determines whether the option is a call or a put.
 CALPUT = 'C'
 A call. The holder has a right to buy.
 CALPUT = 'P'
 A put. The holder has a right to sell.
Constraint: CALPUT = 'C' or 'P'.
- 2: TYPE – CHARACTER*2 *Input*
On entry: indicates the barrier type as **In** or **Out** and its relation to the price of the underlying asset as **Up** or **Down**.
 TYPE = 'DI'
 Down-and-In.
 TYPE = 'DO'
 Down-and-Out.
 TYPE = 'UI'
 Up-and-In.
 TYPE = 'UO'
 Up-and-Out.
Constraint: TYPE = 'DI', 'DO', 'UI' or 'UO'.
- 3: M – INTEGER *Input*
On entry: the number of strike prices to be used.
Constraint: $M \geq 1$.
- 4: N – INTEGER *Input*
On entry: the number of times to expiry to be used.
Constraint: $N \geq 1$.

- 5: $X(M)$ – **double precision** array *Input*
On entry: $X(i)$ must contain X_i , the i th strike price, for $i = 1, 2, \dots, M$.
Constraint: $X(i) \geq z$ and $X(i) \leq 1/z$, where $z = X02AMF()$, the safe range parameter, for $i = 1, 2, \dots, M$.
- 6: S – **double precision** *Input*
On entry: S , the price of the underlying asset.
Constraint: $S \geq z$ and $S \leq 1/z$, where $z = X02AMF()$, the safe range parameter.
- 7: H – **double precision** *Input*
On entry: the barrier price.
Constraint: $H \geq z$ and $H \leq 1/z$, where $z = X02AMF()$, the safe range parameter.
- 8: K – **double precision** *Input*
On entry: the value of a possible cash rebate to be paid if the option has not been knocked in (or out) before expiration.
Constraint: $K \geq 0.0$.
- 9: $T(N)$ – **double precision** array *Input*
On entry: $T(i)$ must contain T_i , the i th time, in years, to expiry, for $i = 1, 2, \dots, N$.
Constraint: $T(i) \geq z$, where $z = X02AMF()$, the safe range parameter, for $i = 1, 2, \dots, N$.
- 10: $SIGMA$ – **double precision** *Input*
On entry: σ , the volatility of the underlying asset. Note that a rate of 15% should be entered as 0.15.
Constraint: $SIGMA > 0.0$.
- 11: R – **double precision** *Input*
On entry: r , the annual risk-free interest rate, continuously compounded. Note that a rate of 5% should be entered as 0.05.
Constraint: $R \geq 0.0$.
- 12: Q – **double precision** *Input*
On entry: q , the annual continuous yield rate. Note that a rate of 8% should be entered as 0.08.
Constraint: $Q \geq 0.0$.
- 13: $P(LDP, N)$ – **double precision** array *Output*
On exit: the $m \times n$ array P contains the computed option prices.
- 14: LDP – INTEGER *Input*
On entry: the first dimension of the array P as declared in the (sub)program from which S30FAF is called.
Constraint: $LDP \geq M$.
- 15: $IFAIL$ – INTEGER *Input/Output*
On entry: $IFAIL$ must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
On exit: $IFAIL = 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, $CALPUT \neq 'C'$ or $'P'$.

$IFAIL = 2$

On entry, $TYPE \neq 'DI', 'DO', 'UI'$ or $'UO'$.

$IFAIL = 3$

On entry, $M \leq 0$.

$IFAIL = 4$

On entry, $N \leq 0$.

$IFAIL = 5$

On entry, $X(i) < z$ or $X(i) > 1/z$, where $z = X02AMF()$, the safe range parameter.

$IFAIL = 6$

On entry, $S < z$ or $S > 1/z$, where $z = X02AMF()$, the safe range parameter.

$IFAIL = 7$

On entry, $H < z$ or $H > 1/z$, where $z = X02AMF()$, the safe range parameter.

$IFAIL = 8$

On entry, $K < 0$.

$IFAIL = 9$

On entry, $T(i) < z$, where $z = X02AMF()$, the safe range parameter.

$IFAIL = 10$

On entry, $SIGMA \leq 0.0$.

$IFAIL = 11$

On entry, $R < 0.0$.

$IFAIL = 12$

On entry, $Q < 0.0$.

$IFAIL = 15$

S and H are not consistent with $TYPE$.

IFAIL = 14

On entry, $LDP < M$.

7 Accuracy

The accuracy of the output is dependent on the accuracy of the cumulative Normal distribution function, Φ . This is evaluated using a rational Chebyshev expansion, chosen so that the maximum relative error in the expansion is of the order of the *machine precision* (see S15ABF and S15ADF). An accuracy close to *machine precision* can generally be expected.

8 Further Comments

None.

9 Example

This example computes the price of a Down-and-In put with a time to expiry of 6 months, a stock price of 100 and a strike price of 100. The barrier value is 95 and there is a cash rebate of 3, payable on expiry if the option has not been knocked in. The risk-free interest rate is 8% per year, there is an annual dividend return of 4% and the volatility is 30% per year.

9.1 Program Text

```
*      S30FAF Example Program Text
*      Mark 22 Release. NAG Copyright 2007.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          LDP, MMAX, NMAX
      PARAMETER        (LDP=8,MMAX=8,NMAX=8)
*      .. Local Scalars ..
      DOUBLE PRECISION H, K, Q, R, S, SIGMA
      INTEGER          I, IFAIL, J, M, N
      CHARACTER        PUT
      CHARACTER*2      TYPE
*      .. Local Arrays ..
      DOUBLE PRECISION P(LDP,NMAX), T(NMAX), X(MMAX)
*      .. External Subroutines ..
      EXTERNAL         S30FAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'S30FAF Example Program Results'
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Standard Barrier Option'
*      Skip heading in data file
      READ (NIN,*)
*      Read problem parameters., TYPE
      READ (NIN,*) PUT, TYPE
      READ (NIN,*) S, H, K, SIGMA, R, Q
      READ (NIN,*) M, N
*
      IF (M.LE.MMAX .AND. N.LE.NMAX) THEN
*      Read array of strike/exercise prices, X
      READ (NIN,*) (X(I),I=1,M)
*      Read array of times to expiry
      READ (NIN,*) (T(I),I=1,N)
*
      IFAIL = 1
*
      CALL S30FAF(PUT,TYPE,M,N,X,S,H,K,T,SIGMA,R,Q,P,LDP,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
        IF (PUT.EQ.'C' .OR. PUT.EQ.'c') THEN
          WRITE (NOUT,*) 'Call :'
        ELSE IF (PUT.EQ.'P' .OR. PUT.EQ.'p') THEN
          WRITE (NOUT,*) 'Put :'
```

```

      END IF
      IF (TYPE.EQ.'DI' .OR. TYPE.EQ.'di' .OR. TYPE.EQ.'Di' .OR.
+       TYPE.EQ.'dI') THEN
        WRITE (NOUT,*) 'Down-and-In'
      ELSE IF (TYPE.EQ.'DO' .OR. TYPE.EQ.'do' .OR. TYPE.EQ.
+       'Do' .OR. TYPE.EQ.'dO') THEN
        WRITE (NOUT,*) 'Down-and-Out'
      ELSE IF (TYPE.EQ.'UI' .OR. TYPE.EQ.'ui' .OR. TYPE.EQ.
+       'Ui' .OR. TYPE.EQ.'uI') THEN
        WRITE (NOUT,*) 'Up-and-In'
      ELSE IF (TYPE.EQ.'UO' .OR. TYPE.EQ.'uo' .OR. TYPE.EQ.
+       'Uo' .OR. TYPE.EQ.'uO') THEN
        WRITE (NOUT,*) 'Up-and-Out'
      END IF
*
      WRITE (NOUT, '(A,1X,F8.4)') ' Spot      = ', S
      WRITE (NOUT, '(A,1X,F8.4)') ' Barrier   = ', H
      WRITE (NOUT, '(A,1X,F8.4)') ' Rebate    = ', K
      WRITE (NOUT, '(A,1X,F8.4)') ' Volatility = ', SIGMA
      WRITE (NOUT, '(A,1X,F8.4)') ' Rate      = ', R
      WRITE (NOUT, '(A,1X,F8.4)') ' Dividend  = ', Q
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Strike      Expiry      Option Price'
      DO 40 I = 1, M
        DO 20 J = 1, N
          WRITE (NOUT,99999) X(I), T(J), P(I,J)
20      CONTINUE
40      CONTINUE
      ELSE
        WRITE (NOUT,*)
        WRITE (NOUT,99998) IFAIL
      END IF
    END IF
*
99999 FORMAT (1X,2(F9.4,1X),6X,F9.4)
99998 FORMAT (1X,' ** S30FAF returned with IFAIL = ',I5)
END

```

9.2 Program Data

S30FAF Example Program Data

```

'P' 'DI' : Call = 'C', Put = 'P', Type
100.0 95.0 3.0 0.3 0.08 0.04 : S, H, K, SIGMA, R, Q
1 1 : M, N
100.0 : X(I), I = 1,2,...M
0.5 : T(I), I = 1,2,...N

```

9.3 Program Results

S30FAF Example Program Results

Standard Barrier Option

Put :

Down-and-In

```

Spot      = 100.0000
Barrier    = 95.0000
Rebate     = 3.0000
Volatility = 0.3000
Rate       = 0.0800
Dividend   = 0.0400

```

```

Strike      Expiry      Option Price
100.0000    0.5000      7.7988

```